
Introduction
    

Topical is a library for defining and serving topics. A topic is a value that can be efficiently
observed in real-time by connected clients. The server-side implementation takes care of defining
how to intialise the state, and then how to keep it updated.
Multiple clients can subscribe to an instance of a topic - instances are identifier by a path,
which matches the route defined by a topic. Multiple instances of a topic can exist by using route
placeholders.
Topical takes care of starting topic instances as needed, and stopping them once all clients have
disconnected.
Uses
Topics are implemented in Elixir, and behave somewhat like GenServers - the main difference being
that their state is easily observable by clients. The main benefit comes from using the JavaScript
(and React) client (and a WebSocket adapter) to observe the state, but it's also possible to use the
Elixir API.
There's a certain amount of flexibility in how topics are implemented. They can be be used to track
ephemeral state - for example user presence, or cursor positions. They can be backed by simple
mechanisms like a file or file-based database. Or they can sit in front of an RDBMS (e.g., utilising
Postgres notification channels), providing real-time cached views. Or they can be used to implement
the event sourcing pattern.
Another use case is for implementing an incremental cache: the state of the cache is setup in the topic
initalisation and then kept up to date as necessary, without having to refresh from scratch. Clients
accessing the cache will be able to read up-to-date state directly from memory.



  

    
Getting started
    

First you need to implement a topic definition. For example, a todo list topic might track todo
items and their order, handle requests from clients to add new items or update the text of existing
items, and handle Erlang messages from a separate process indicating when items are done:
defmodule MyApp.Topics.List do
  use Topical.Topic, route: ["lists", :list_id]

  # Initialise the topic
  def init(params) do
    list_id = Keyword.fetch!(params, :list_id)

    value = %{items: %{}, order: []} # exposed 'value' of the topic
    state = %{list_id: list_id, last_item_id: 0} # hidden server state
    topic = Topic.new(value, state)

    {:ok, topic}
  end

  # Optionally, handle execution of an action
  def handle_execute("add_item", {text}, topic, _context) do
    {id, topic} = generate_item_id(topic)

    # Update the topic by putting the item in 'items', and appending the id to 'order'
    topic =
      topic
      |> Topic.set([:items, id], %{text: text, done: false})
      |> Topic.insert([:order], id)

    # Return the result (the 'id'), and the updated topic
    {:ok, id, topic}
  end

  defp generate_item_id(topic) do
    id = topic.state.last_item_id + 1
    topic = put_in(topic[:state][:last_item_id], id)
    {Integer.to_string(id), topic}
  end

  # Optionally, handle a notification (an action without a result)
  def handle_notify("update_text", {id, text}, topic) do
    topic  = Topic.set(topic, [:items, id, :text], text)
    {:ok, topic}
  end

  # Optionally, handle Erlang messages
  def handle_info({:done, id}, topic) do
    topic  = Topic.set(topic, [:items, id, :done], true)
    {:ok, topic}
  end
end
Then add a Topical registry to your application supervision tree, referencing the topic:
defmodule MyApp.Application do
  use Application

  @impl true
  def start(_type, _args) do
    children = [
      # ...
      {Topical, name: MyApp.Topical, topics: [MyApp.Topics.List, ...]},
    ]

    Supervisor.start_link(children, ...)
  end
end
At this point you should be able to subscribe to the topic:
{:ok, ref} = Topical.subscribe(MyApp.Topical, "lists/foo", self())
After subscribing, the process will be sent an initial {:reset, ref, value} message (where ref
is the subscription reference returned from subscribe), and then subsequent
{:updates, ref, updates} messages, where updates is a list of updates, each taking the form:
	{:set, path, value}: the value has been set at the path.
	{:unset, path, key}: the key has been unset from the object at the path.
	{:insert, path, index, values}: the values have been inserted into the array at the path.
	{:delete, path, index, count}: count values have been deleted from the array at the path, from the position index.
	{:merge, path, value}: the value has been (shallow) merged at the path.

(You can receive any waiting messages on an IEx shell with:
receive do x -> x after 0 -> nil end.)
To execute an action:
{:ok, item_id} = Topical.execute(MyApp.Topical, "lists/foo", "add_item", {"Test item", false})
To unsubscribe:
Topical.unsubscribe(MyApp.Topical, "lists/foo", ref)
However, rather than using the API from Elixir, you may wish to set up a
WebSocket adapter, and then use the JavaScript client.



  

    
WebSocket adapter
    

Rather than interacting with with the Topical API directly, you can expose Topical from your web
server. Separate adapters exist for Cowboy and WebSock (for use with Plug and Bandit).
In both cases, the Topical registry must be specified (Todo.Registry in the examples below).
Cowboy adapter
If you're using Cowboy, the adapter can be added to your routes:
:cowboy_router.compile([
  {:_,
   [
     # ...
     {"/socket", Topical.Adapters.Cowboy.WebsocketHandler, registry: Todo.Registry}
   ]}
])
WebSock adapter
The WebSock adapter is compatible with Plug (and Bandit):
defmodule Router do
  use Plug.Router

  plug :match
  plug :dispatch

  # ...

  get "/socket" do
    conn
    |> WebSockAdapter.upgrade(
      Topical.Adapters.Plug.WebSockServer,
      [registry: Todo.Registry],
      timeout: 60_000
    )
    |> halt()
  end

  # ...
end
Context
Optionally, an init function can be passed, which will be called before the connection is
upgraded. It will be passed the Cowboy request or Plug conn, and must return {:ok, context}. The
context will then be passed to the topic. This can be useful for authentication: for
unauthenticated users, return an error result to prevent the socket getting established; for
authenticated users, include the user ID in the context so it can be used for
authorisation/identification within a topic.
Client
You can now connect to Topical from a JavaScript client.



  

    
JavaScript client
    

You can connect to a Topical server that has been exposed by an adapter, using the JavaScript
client. The client is available on npm:
npm install @topical/core

(Or you may prefer to use the React client - see below.)
type ListModel = {
  items: Record<string, { text: string, done: boolean }>;
  order: string[];
}

// Setup the socket
const socket = new Socket("ws://example.com/socket");

// Subscribe to a topic (returns a function to unsibscribe)
const unsubscribe = socket.subscribe<ListModel>(
  ["lists", "foo"],
  (list: ListModel) => { console.log(value); },
  (error) => { ... }
);

// Execute an action
const itemId = await socket.execute(["lists", "foo"], "add_item", "First item");

// (The subscription should have been updated)

// Send a notification
socket.notify(["lists", "foo"], "update_item", itemId, "Inaugural item")

// (The subscription should have been updated again)

// Unsubscribe
unsubscribe();
React client
Instead of using the JavaScript client directly, you can use the React client. Install it from npm:
npm install @topical/react

Setup the socket using the provider:
import { SocketProvider } from "@topical/react";

function getSocketUrl() {
  // TODO
  return "ws://example.com/socket";
}

function App() {
  return (
    <SocketProvider url={getSocketUrl()}>
      // ...
    </SocketProvider>
  );
}
Then use the useTopic hook in your components to subscribe to your topic:
import { useTopic } from "@topical/react";

function List({ id }) {
  const [list, { execute, notify, loading, error }] = useTopic<models.List>("lists", id);
  const addItem = useCallback(
    (text: string) => execute("add_item", text),
    [execute]
  );
  if (loading) {
    return <p>Loading...</p>;
  } else if (error) {
    return <p>Error.</p>
  } else {
    return (
      // ...
    );
  }
}
If you need access to the underlying socket (or the status), you can use the useSocket hook:
import { useSocket } from "@topical/react";

function SocketStatus() {
  const [_socket, state] = useSocket();
  return <p>{state}</p>;
}



  

    
Topical 
    



      
This module provides the high level interface for interacting with topics. Primarily for
subscribing (and unsubscribing), but also for sending requests.
After subscribing, a client will initially receive a {:reset, ref, value} message, and then
subsequent {:updates, ref, updates} messages when the value of the topic changes, where
updates is a list with each item being one of:
	{:set, path, value}: the value has been set at the path.
	{:unset, path, key}: the key has been unset from the object at the path.
	{:insert, path, index, values}: the values have been inserted into the array at the path.
	{:delete, path, index, count}: count values have been deleted from the array at the path, from the position index.

A client can interact directly with a topic by executing actions (which returns a result), or
by notifying (without waiting for a result). These are analogous to GenServer.call/3 and
GenServer.cast/2. Be aware that a topic is blocked while processing a request.

    Functions
  


    
      
        capture(registry, topic, context \\ nil)

      


        Captures the state of the topic (in the specified registry) without subscribing.



    


    
      
        child_spec(options)

      


        Returns a specification to start a Topical registry under a supervisor.



    


    
      
        execute(registry, topic, action, args \\ {}, context \\ nil)

      


        Executes an action in a topic.



    


    
      
        notify(registry, topic, action, args \\ {}, context \\ nil)

      


        Send a notification to a registry.



    


    
      
        subscribe(registry, topic, pid, context \\ nil)

      


        Subscribes to the specified topic (in the specified registry).



    


    
      
        unsubscribe(registry, topic, ref)

      


        Unsubscribes from a topic (in the specified registry).

      Link to this function
    
    capture(registry, topic, context \\ nil)



  


  

Captures the state of the topic (in the specified registry) without subscribing.

  
  example

  
  Example


Topical.capture(MyApp.Topical, ["lists", "foo"])
# => {:ok, %{items: %{}, order: []}}

  



  
    
      
      Link to this function
    
    child_spec(options)



  


  

Returns a specification to start a Topical registry under a supervisor.

  



    

    

  
    
      
      Link to this function
    
    execute(registry, topic, action, args \\ {}, context \\ nil)



  


  

Executes an action in a topic.

  
  example

  
  Example


Topical.execute(MyApp.Topical, ["lists", "foo"], "add_item", {"Test", false})
#=> {:ok, "item123"}

  



    

    

  
    
      
      Link to this function
    
    notify(registry, topic, action, args \\ {}, context \\ nil)



  


  

Send a notification to a registry.
This is similar to execute/4, except no result is waited for.

  
  example

  
  Example


Topical.notify(MyApp.Topical, ["lists", "foo"], "update_done", {"item123", true})
#=> :ok

  



    

  
    
      
      Link to this function
    
    subscribe(registry, topic, pid, context \\ nil)



  


  

Subscribes to the specified topic (in the specified registry).
Returns {:ok, ref}, where the ref is a reference to the subscription.
The pid will be send messages, as described above.

  
  example

  
  Example


Topical.subscribe(MyApp.Topical, ["lists", "foo"], self())
#=> {:ok, #Reference<0.4021726225.4145020932.239110>}

  



  
    
      
      Link to this function
    
    unsubscribe(registry, topic, ref)



  


  

Unsubscribes from a topic (in the specified registry).

  
  example

  
  Example


Topical.unsubscribe(MyApp.Topical, ["lists", "foo"], ref)

  


        

      



  

    
Topical.Adapters.Cowboy.RestHandler 
    



      
A REST-ish handler adapter for a Cowboy web server.
Options
	registry - The name of the Topical registry. Required.
	init - A function called before starting/capturing the topic, passed the request. The
function must return {:ok, context} for the connection to be accepted. This context is
then passed to topics.

Example
:cowboy_router.compile([
  {:_,
   [
     # ...
     {"/topics/[...]", RestHandler, registry: MyApp.Topical}
   ]}
])

      





  

    
Topical.Adapters.Cowboy.WebsocketHandler 
    



      
A WebSocket handler adapter for a Cowboy web server.
Options
	registry - The name of the Topical registry. Required.
	init - A function called before upgrading the connection, which is passed the request. The
function must return {:ok, context} for the connection to be accepted. This context is then
passed to topics.

Example
:cowboy_router.compile([
  {:_,
   [
     # ...
     {"/socket", WebsocketHandler, registry: MyApp.Topical}
   ]}
])

      





  

    
Topical.Adapters.Plug.WebSockServer 
    



      
A WebSocket server for a WebSock service (for use with Plug).
Options
	registry - The name of the Topical registry. Required.
	init - A function called before upgrading the connection, which is passed the request. The
function must return {:ok, context} for the connection to be accepted. This context is then
passed to topics.

Example
defmodule Router do
  use Plug.Router

  plug :match
  plug :dispatch

  # ...

  get "/socket" do
    conn
    |> WebSockAdapter.upgrade(
      Topical.Adapters.Plug.WebSockServer,
      [registry: MyRegistry],
      timeout: 60_000)
    |> halt()
  end

  # ...
end

    Functions
  


    
      
        handle_in(arg, state)

      


    


    
      
        handle_info(info, state)

      


    


    
      
        init(opts)

      Link to this function
    
    handle_in(arg, state)



  


  


  



  
    
      
      Link to this function
    
    handle_info(info, state)



  


  


  



  
    
      
      Link to this function
    
    init(opts)



  


  


  


        

      



  

    
Topical.Topic 
    



      
This module provides functions for instantiating and manipulating topic state.
The state of a topic is composed of a value, observed by subscribed clients, and also hidden
internal state, which can be used to share state between calls. The value must only be
manipulated by the helper functions (which track the individual updates). The hidden state can
be modified directly.
This module also contains a macro - use-ing it sets up a topic server:
defmodule MyApp.Topics.List do
  use Topical.Topic, route: ["lists", :list_id]

  # Initialise the topic
  def init(params) do
    list_id = Keyword.fetch!(params, :list_id)

    value = %{items: %{}, order: []} # exposed 'value' of the topic
    state = %{list_id: list_id} # hidden server state
    topic = Topic.new(value, state)

    {:ok, topic}
  end

  # Optionally, handle subscribe
  def handle_subscribe(topic, _context) do
    {:ok, topic}
  end

  # Optionally, handle unsubscribe
  def handle_unsubscribe(topic, _context) do
    {:ok, topic}
  end

  # Optionally, handle capture
  def handle_capture(topic, _context) do
    {:ok, topic}
  end

  # Optionally, handle execution of an action
  def handle_execute("add_item", {text}, topic, _context) do
    id = Integer.to_string(:erlang.system_time())

    # Update the topic by putting the item in 'items', and appending the id to 'order'
    topic =
      topic
      |> Topic.set([:items, id], %{text: text, done: false})
      |> Topic.insert([:order], id)

    # Return the result (the 'id'), and the updated topic
    {:ok, id, topic}
  end

  # Optionally, handle a notification (an action without a result)
  def handle_notify("update_text", {id, text}, topic, _context) do
    topic  = Topic.set(topic, [:items, id, :text], text)
    {:ok, topic}
  end

  # Optionally, handle Erlang messages
  def handle_info({:done, id}, topic) do
    topic  = Topic.set(topic, [:items, id, :done], true)
    {:ok, topic}
  end

  # Optionally, handle the topic being terminated (e.g., once clients have disconnected)
  def terminate(_reason, topic) do
    # ...
  end
end

    Functions
  


    
      
        delete(topic, path, index, count \\ 1)

      


        Updates the topic be deleting count values from the array at the path, from the index.



    


    
      
        insert(topic, path, index \\ nil, value)

      


        Updates the topic by inserting the specified values into the array at the path, at the
index (or append them to the end, if no index is specified).



    


    
      
        merge(topic, path, value)

      


        Updates the opic by merging value into path.



    


    
      
        new(value, state \\ nil)

      


        Instantiates a new instance of a topic.



    


    
      
        set(topic, path, value)

      


        Updates the topic by setting the value at the path.



    


    
      
        unset(topic, path, key)

      


        Updates the topic by unsetting the key of the object at the path.

      Link to this function
    
    delete(topic, path, index, count \\ 1)



  


  

Updates the topic be deleting count values from the array at the path, from the index.
%{foo: %{bar: [1, 2, 3, 4]}}
|> Topic.new()
|> Topic.delete([:foo, :bar], 2)
|> Map.fetch!(:value)
#=> %{foo: %{bar: [1, 2, 4]}}

  



    

  
    
      
      Link to this function
    
    insert(topic, path, index \\ nil, value)



  


  

Updates the topic by inserting the specified values into the array at the path, at the
index (or append them to the end, if no index is specified).
If value is a list, all the items will be added (to add a list as a single item, wrap it in a
list).
%{foo: %{bar: [1, 4]}}
|> Topic.new()
|> Topic.insert([:foo, :bar], 1, [2, 3])
|> Map.fetch!(:value)
#=> %{foo: %{bar: [1, 2, 3, 4]}}

  



  
    
      
      Link to this function
    
    merge(topic, path, value)



  


  

Updates the opic by merging value into path.
%{foo: %{bar: %{a: 1, b: 2}}}
|> Topic.new()
|> Topic.merge([:foo, :bar], %{b: 3, c: 4})
|> Map.fetch!(:value)
#=> %{foo: %{bar: %{a: 1, b: 3, c: 4}}}

  



    

  
    
      
      Link to this function
    
    new(value, state \\ nil)



  


  

Instantiates a new instance of a topic.
value is the initial value of the client-visible state. state is the 'hidden' internal state.

  



  
    
      
      Link to this function
    
    set(topic, path, value)



  


  

Updates the topic by setting the value at the path.
%{foo: %{bar: 2}}
|> Topic.new()
|> Topic.set([:foo, :bar], 3)
|> Map.fetch!(:value)
#=> %{foo: %{bar: 3}}

  



  
    
      
      Link to this function
    
    unset(topic, path, key)



  


  

Updates the topic by unsetting the key of the object at the path.
%{foo: %{bar: 2}}
|> Topic.new()
|> Topic.unset([:foo], :bar)
|> Map.fetch!(:value)
#=> %{foo: %{}}

  


        

      



  

    
Topical.Topic.Diff 

    Functions
  


    
      
        diff(old, new)

      


    


    
      
        diff(old, new, path)

      Link to this function
    
    diff(old, new)



  


  


  



  
    
      
      Link to this function
    
    diff(old, new, path)



  


  


  


        

      



  

    
Topical.Topic.Server behaviour
    



      
This module defines the bahaviour of a topic.
See Topical.Topic for a usage example.

    Callbacks
  


    
      
        handle_capture(topic, context)

      


        Invoked before state is captured (after initialisation).



    


    
      
        handle_execute(action, args, topic, context)

      


        Invoked when a client has executed an action.



    


    
      
        handle_info(msg, topic)

      


        Invoked to handle other messages.



    


    
      
        handle_notify(action, args, topic, context)

      


        Invoked when a client has sent a notification.



    


    
      
        handle_subscribe(topic, context)

      


        Invoked before a client subscribes (but after initialisation).



    


    
      
        handle_unsubscribe(topic, context)

      


        Invoked after a client unsubscribes (either explicitly or because the process dies).



    


    
      
        init(params)

      


        Invoked when the topic is started to get the initial state.



    


    
      
        terminate(reason, topic)

      


        Invoked when a topic has been stopped.



    





  
    Functions
  


    
      
        child_spec(init_arg)

      


        Returns a specification to start this module under a supervisor.



    


    
      
        start_link(options)

      


        Starts a topic server process linked to the current process.

      Link to this callback
    
    handle_capture(topic, context)



  


  

      

          @callback handle_capture(
  topic :: %Topical.Topic{state: term(), updates: term(), value: term()},
  context :: any()
) :: {:ok, %Topical.Topic{state: term(), updates: term(), value: term()}}


      


Invoked before state is captured (after initialisation).
This callback is optional.

  



  
    
      
      Link to this callback
    
    handle_execute(action, args, topic, context)



  


  

      

          @callback handle_execute(
  action :: term(),
  args :: tuple(),
  topic :: %Topical.Topic{state: term(), updates: term(), value: term()},
  context :: any()
) ::
  {:ok, term(), %Topical.Topic{state: term(), updates: term(), value: term()}}


      


Invoked when a client has executed an action.
This callback is optional. If one is not implemented, the topic will fail if an action is
executed.

  



  
    
      
      Link to this callback
    
    handle_info(msg, topic)



  


  

      

          @callback handle_info(
  msg :: term(),
  topic :: %Topical.Topic{state: term(), updates: term(), value: term()}
) ::
  {:ok, %Topical.Topic{state: term(), updates: term(), value: term()}}


      


Invoked to handle other messages.
This callback is optional.

  



  
    
      
      Link to this callback
    
    handle_notify(action, args, topic, context)



  


  

      

          @callback handle_notify(
  action :: term(),
  args :: tuple(),
  topic :: %Topical.Topic{state: term(), updates: term(), value: term()},
  context :: any()
) :: {:ok, %Topical.Topic{state: term(), updates: term(), value: term()}}


      


Invoked when a client has sent a notification.
This callback is optional. If one is not implemented, the topic will fail if a notification is
received.

  



  
    
      
      Link to this callback
    
    handle_subscribe(topic, context)



  


  

      

          @callback handle_subscribe(
  topic :: %Topical.Topic{state: term(), updates: term(), value: term()},
  context :: any()
) :: {:ok, %Topical.Topic{state: term(), updates: term(), value: term()}}


      


Invoked before a client subscribes (but after initialisation).
This callback can be used to update the topic, for example (in combination with
handle_unsubscribe) to track connected users.
This callback is optional.

  



  
    
      
      Link to this callback
    
    handle_unsubscribe(topic, context)



  


  

      

          @callback handle_unsubscribe(
  topic :: %Topical.Topic{state: term(), updates: term(), value: term()},
  context :: any()
) :: {:ok, %Topical.Topic{state: term(), updates: term(), value: term()}}


      


Invoked after a client unsubscribes (either explicitly or because the process dies).
This callback is optional.

  



  
    
      
      Link to this callback
    
    init(params)



  


  

      

          @callback init(params :: [...]) ::
  {:ok, %Topical.Topic{state: term(), updates: term(), value: term()}}
  | {:error, reason :: any()}


      


Invoked when the topic is started to get the initial state.
params are the values associated with the placeholders in the route.

  



  
    
      
      Link to this callback
    
    terminate(reason, topic)



  


  

      

          @callback terminate(
  reason :: term(),
  topic :: %Topical.Topic{state: term(), updates: term(), value: term()}
) :: term()


      


Invoked when a topic has been stopped.
This callback is optional.

      Link to this function
    
    child_spec(init_arg)



  


  

Returns a specification to start this module under a supervisor.
See Supervisor.

  



  
    
      
      Link to this function
    
    start_link(options)



  


  

Starts a topic server process linked to the current process.

  
  options

  
  Options


	:module - the module that implements the topic behaviour.
	:init_arg - the argument passed to the topic's init callback.

